

# Pre-Lab, Week 6: Curves, Surfaces and Maple

In this document, we discuss curves, surfaces and how we can use Maple to analyze them.

## Curves

During our first lab, we talked about **curves (or paths)** in the plane. We can define a curve in the plane as being the **graph** of  $y = f(x)$ . That is, one way of defining a curve  $C$  is formally:

$$C = \{(x, y) \mid y = f(x), \quad a \leq x \leq b\}$$

This definition is very restrictive however- It assumes that your curve  $C$  must be a function of  $x$ . Typically, a general curve should be defined as the image (in the plane or in 3-d) of  $a \leq t \leq b$ , or we say that  $x, y$  have been parameterized in  $t$ . For example,

$$C = \{(x, y) \mid x = f(t), y = g(t), \quad a \leq t \leq b\}$$

Notice that this is very general and does not assume that  $y$  must be a function of  $x$ . The difference between these can also be clearly demonstrated graphically. In Figure below, we graph on the left represents  $y$  as a function of  $x$ . On the right, both  $x$  and  $y$  are functions of  $t$ .




Figure 1: Curves in the plane. On the left,  $y = f(x)$ . On the right, the curve has been parameterized in  $t$ . That is,  $x(t) = \sin(2t)$  and  $y(t) = \cos(\sqrt{2}t)$ . The curve on the right cannot be expressed as  $y = f(x)$ , and is a more general way of defining a curve (or path).

## Maple and Curves

When plotting curves in Maple, we should remember that Maple distinguishes between expressions and functions. We have always plotted using expressions, so we'll show that again here.

To plot  $y$  as a function of  $x$  (in Figure ,  $y = \cos(x/2) + \sin(2x)$ ), we might write:

```
F:=cos(x/2)+sin(2*x);
plot(F, x=0..7);
```

To plot  $x$  and  $y$  as functions of  $t$ , like in Figure , we would write something like:

```
F:=sin(2*t); G:=cos(sqrt(2)*t);
plot( [F,G,t=0..20]);
```

To plot two functions together using  $t$  as the horizontal axis, we would type:

```
plot([F,G],t=0..20);
```

If we want two parametric sets of functions, for example, the previous graph together with the unit circle, we might write it as:

```
plot([ [F,G,t=0..20],[cos(t), sin(t),t=0..2*Pi] ]);
```

## Implicitly Defined Curves

In Calculus II and III, we have treated expressions like:  $f(x, y) = k$  (with  $k$  a constant) as defining  $y$  in terms of  $x$  *implicitly*. For example,


$$\cos(xy) = \frac{1}{3}$$

gives the set of points in the plane shown in Maple by:

```
with(plots):
implicit plot(cos(x*y)=1/3,x=-4..4,y=-4..4);
```

If you run this command, you may notice that some of the curves are not smooth. We can fix this by using a finer grid of points for the plot:

```
implicit plot(cos(x*y)=1/3,x=-4..4,y=-4..4, grid=[90,90]);
```




## Curves in 3-d

We can parameterize a curve in three dimensions simply by adding a third coordinate that depends on  $t$ . That is, a curve  $C$  in 3-d is defined as:

$$C = \{(x, y, z) \mid x = f(t), y = g(t), z = h(t) \quad a \leq t \leq b\}$$

Maple needs a new command for curves in three dimensions- It refers to them as **space curves**. Here's an example, where we plot a curve called a **knot**. In particular, notice that we need `with(plots)` at the top of the worksheet.

```
with(plots):  
F:=-10*cos(t)-2*cos(5*t)+15*sin(2*t);  
G:=-15*cos(2*t)+10*sin(t)-2*sin(5*t);  
H:=10*cos(3*t);  
spacecurve( [F,G,H, t=0..2*Pi] );
```



## Surfaces

We usually start discussing surfaces in Calculus III using the graph of  $z = f(x, y)$ . That is, our  $z$  value is a function of  $(x, y)$ . This is very restrictive in the sense that there are many surfaces (like many curves) that cannot be written as a function of  $(x, y)$ . For example, the surface of a sphere has two  $z$  values for each  $x, y$ :

$$x^2 + y^2 + z^2 = 1 \quad \Rightarrow \quad z = \pm\sqrt{1 - x^2 - y^2}$$

and so the surface of a sphere is not a function of  $(x, y)$  (although we could break it up into a union of two functions). A surface more generally can be thought of (parametrically) as the image of a mapping of the plane into three dimensions. That is,  $x$ ,  $y$ , and  $z$  can each be a function of two other variables  $(u, v)$ . In brief, that means our function performs the following:

$$(u, v) \rightarrow (x(u, v), y(u, v), z(u, v))$$

which we might write in the following way:

$$\mathbf{F}(u, v) = \langle x(u, v), y(u, v), z(u, v) \rangle$$

## Surfaces in Maple

The Maple commands we will look at include:

`plot3d`   `implicitplot3d`   `animate`

Just like for curves, we can either define the graph  $z = f(x, y)$  or in parametric form. Here are three ways of plotting the surface of a sphere. Theoretically, they all produce the same thing, but they differ in how the sphere is defined.

- Plot the upper hemisphere of a sphere of radius 1.

The equation is  $x^2 + y^2 + z^2 = 1$ , we can solve for  $z$  and take the positive root:

$$z = \sqrt{1 - x^2 - y^2}$$

In the plot command, there are several extra options- See if you can figure out what they do (that is, `axes=normal` (or you can use `axes=boxed` and the `scaling=constrained` is by default “unconstrained”). The line below is broken for readability- Don’t break the line in Maple.

```
with(plots):  
plot3d(sqrt(1-x^2-y^2), x=-1.5..1.5, y=-1.5..1.5,  
        axes=NORMAL, scaling=CONSTRAINED);
```

- Just as before, the surface may be defined implicitly. For example, the sphere is  $x^2 + y^2 + z^2 = 1$ , and we might plot it directly:

```
implicitplot3d(x^2+y^2+z^2=1, x=-2..2, y=-2..2, z=-2..2);
```

It looks more like a soccer ball than a sphere- You can get a better plot by inserting the option: `numpoints=1500`

- You can also plot the surface of a sphere parametrically, which you may not have seen before. We define the functions for  $x, y$  and  $z$  as  $F, G$  and  $H$ :

```
F:=cos(u)*sin(v);  
G:=sin(u)*sin(v);  
H:=cos(v);  
plot3d([F,G,H], u=0..2*Pi, v=0..Pi);
```

Here are a couple of extra plots with some extra options, like color and animation:

- Plot the surface  $z = xe^{-x^2-y^2}$ , and color the graph according to  $\sin(xy)$ :

```
plot3d(x*exp(-x^2-y^2), x = -2..2, y = -2..2, color = sin(x*y))
```

Just for fun, we can also do an animation, where we vary the height of the valley and hilltop by using an extra parameter  $A$ :

```
animate( plot3d, [sin(A)*x*exp(-x^2-y^2), x=-2..2, y=-2..2],
          A=0..2*Pi, shading=zhue );
```

Then click on the graph and there will be a play button on the menu. There is also an option to run a loop- See if you can find it.

## Multiple Surfaces

Just as with curves, we can plot multiple surfaces together. Two ways are shown below:

*Method 1:* Plot the two separately, then “display” the results. Use this technique if you want to overlay graphs over the top of each other (or with each other).

In this example, plot  $z = \sin(x)\sin(y)$ , then plot the parametrized half sphere (as before).

```
with(plots):
A:=plot3d(sin(x)*sin(y),x=-Pi..Pi,y=-Pi..Pi,color=white):
B:=plot3d(x*exp(-x^2-y^2), x = -2..2, y = -2..2, color = sin(x*y));
display({A,B});
```

*Method 2:* If your functions are all of the form  $z = f(x, y)$ , you can use a single plot command. For example, here we plot three surfaces:  $\sin(xy)$ ,  $x - y$  and  $4e^{-x^2-y^2}$ , all on the same graph:

```
plot3d({sin(x*y), x-y, 4*exp(-x^2-y^2)},x=-2..2,y=-2..2);
```

*Question:* What happens if you use square brackets around the functions instead of curly braces?... Does the same thing happen with 4 functions?

## Level Curves

Also of interest are the level curves of a function. Maple uses the `contourplot` or `contourplot3d` function. For example, we plot the contours for  $z = \sqrt{x^2 - y^2}$  for contours  $z = 1/2, 3, 4$  by giving the following command (assumes you have already given the `with(plots)` command).

```
contourplot(sqrt(x^2-y^2),x=-5..5,y=-5..5,
            contours=[1/2,3,4],grid=[100,100]);
```

As a second example, here we use the three-dimensional version of the contour plot, and we use an extra option that helps with the visualization. This surface is called a “dog saddle” (a dog has 4 legs, usually):

```
contourplot3d(x*y*(x^2-y^2),x=-3..3,y=-3..3,filledregions = true);
```

## Multivariate Limits

Multivariate limits can be difficult to compute by hand- Maple can have a difficult time as well. However, often Maple is able to compute a multivariate limit- and we use the `limit` command we learned earlier. There are some excellent examples in the help file: `?limit`

### Examples

1. Find  $\lim_{(x,y) \rightarrow (0,0)} \frac{3x^2y}{x^2 + y^2}$ , if it exists.

You should look at the plot first to see if you can first guess.

SOLUTION in Maple:

```
F:=3*x^2*y/(x^2+y^2);  
plot3d(F,x=-1..1,y=-1..1)
```

We see a nice surface- It may have some edges, but nothing that should give a discontinuity. See if Maple can find the limit:

```
limit(F,{x=0,y=0});
```

Maple's having a hard time here... Using some algebra, we see that:

$$x^2 \leq x^2 + y^2 \quad \Rightarrow \quad \frac{x^2}{x^2 + y^2} \leq 1 \quad \Rightarrow \quad \frac{3x^2y}{x^2 + y^2} \leq 3|y|$$

And with a similar argument, the function is bounded below by  $-3|y|$ , which we can verify with a plot:

```
plot3d({-3*abs(y),F,3*abs(y)},x=-1..1,y=-3..3);
```

Therefore, by the Squeeze Theorem, the limit is 0.

2. Show that the following limit either exists or show that it does not exist:

$$\lim_{(x,y) \rightarrow (0,0)} \frac{y^2 \sin^2(x)}{x^4 + y^4}$$

SOLUTION: First plot it to see what it looks like:

```
G:=y^2*(sin(x))^2/(x^4+y^4);  
plot3d(G,x=-1..1,y=-1..1,shading=zhue,axes=boxed);
```

If you go to the origin along a purple “valley”, the limit will be zero. This would be, for example, along either the  $x$ - or  $y$ -axis:

$$\lim_{x \rightarrow 0, y} \frac{y^2 \sin^2(x)}{x^4 + y^4} = 0$$

However, trying to go to the origin along the orange “hill”,  $y = x$ , looks like a different limit. Let’s try it: Along  $y = x$ , the expression becomes

$$\frac{x^2 \sin^2(x)}{2x^4} = \frac{\sin^2(x)}{2x^2} = \frac{1}{2} \left( \frac{\sin(x)}{x} \right)^2$$

You may recall that the limit in the parentheses is 1 (l’Hospital’s rule, for example), so that the overall limit is now  $1/2$ - which agrees with our graph.

Therefore, the limit at the origin does not exist, and this function is not continuous at the origin.

This is typical of this kind of discontinuity.

3. Does this limit exist?

$$\lim_{(x,y) \rightarrow (0,0)} \frac{x^4 - 4y^2}{x^2 + 2y^2}$$

SOLUTION: Plot in Maple, and you’ll see something similar to the last example- This limit does not exist.

```
plot3d((x^4-4*y^2)/(x^2+2*y^2),x=-1..1,y=-1..1,shading=zhue,axes=boxed);
```

From the graph, we see that if we travel along the  $y$ -axis, the limit is  $-2$ , but if we travel along the  $x$ -axis, the limit is 0 (we would want to actually show the algebra as well).

## Derivatives and Partial Derivatives

Partial derivatives are easy in Maple, as are higher derivatives.

- **EXAMPLE:** Given  $f(x, y) = \sqrt{x^2 + y^2}$ , find  $f_x$ ,  $f_{yx}$  and  $f_{yyx}$

SOLUTION: (Note that we’re differentiating expressions and not functions)

```
f:=sqrt(x^2+y^2);
fx:=diff(f,x);
fyx:=diff(f,y,x);
fyyx:=diff(f,y,y,x);
```

Is  $f_{yyx}$  the same as  $f_{yxy}$ ? Check in Maple.

- Suppose  $u(x, y) = e^x \sin(y)$ . Show that  $u$  satisfies “Laplace’s Equation”:

$$u_{xx}(x, y) + u_{yy}(x, y) = 0$$

SOLUTION: In Maple, this is easy to check:

```
u:=exp(x)*sin(y);
uxx:=diff(u,x,x);
uyy:=diff(u,y,y);
uxx-uyy;
```

## Maple Exercises (Week 6)

Solve the following in a Maple Worksheet.

1. Plot the  $(4, 3)$ -torus knot in three dimensions, given parametrically by:

$$x(t) = \cos(3t)(3 + \cos(4t)), \quad y(t) = \sin(3t)(3 + \cos(4t)), \quad z(t) = \sin(4t)$$

The knot is a closed curve, meaning that the curve comes together and starts to repeat itself.

Find the arc length of the knot. Hint: You’ll need a starting and ending time. You might start by guessing from the graph. Double check by evaluating  $x, y, z$  at these times (using Maple- Use the **subs** command).

2. (a) Plot the parametric surface in three dimensions:

$$\langle u, \sin(u + v), \sin(v) \rangle \quad -\pi \leq u \leq \pi, -\pi \leq v \leq \pi$$

- (b) Find a parametric representation for the line in  $(u, v)$  coordinates going from  $(-3, 2)$  to  $(1, 3)$ . This will give you two functions,  $u(t)$  and  $v(t)$ . Put this line on the surface in three dimensions as in the last example and plot them together.

- (c) Find the arc length of the curve (in 3d) that you found in the previous part.

3. Consider the surface:

$$\langle u \cos(v), u \sin(v), u \rangle \quad -1 \leq u \leq 1, \quad 0 \leq v \leq 2\pi$$

- (a) Plot the surface in 3d.
- (b) If  $v$  is held constant, what kinds of curves do we get in 3d? Plot two of them (with the surface) all in the same graph.
- (c) If  $u$  is held constant, what kinds of curves do we get in 3d? Plot two of them (with the surface) all in the same graph.



4. Find the limit, if it exists, or show that it does not exist:

(a)  $\lim_{(x,y) \rightarrow (0,0)} \frac{x^2 - 4y^2}{x^2 + 2y^2}$

(b)  $\lim_{(x,y) \rightarrow (0,0)} \frac{2x^2 + 3xy + 4y^2}{3x^2 + 5y^2}$

5. Is the function  $f$  continuous? Why or why not?

$$f(x, y) = \frac{1 + x^2 + y^2}{1 - x^2 - y^2}$$

6. Is the function  $f$  continuous? Why or why not?

$$f(x, y) = \begin{cases} \frac{\sin(xy)}{xy} & xy \neq 0 \\ 1 & xy = 0 \end{cases}$$